BONNES PRATIQUES VBA – Manipuler les formes (shapes)
[bookmark: _Hlk35368646]
MANIPULER LES FORMES (SHAPES) EN VBA


L’un des aspects des plus remarquables d’EXCEL réside dans sa capacité à dessiner une multitude de formes graphiques plus ou moins complexes, de pouvoir gérer leurs tailles, leurs couleurs, leurs positionnements, leurs comportements…
Cette fiche « Bonnes pratiques VBA » présente de façon concise ce qu’il est nécessaire de connaître pour manipuler ces formes par programmation VBA.
(Les copies-écrans du présent document sont issues d’EXCEL version 365)

L’icône  [image: ] signale une remarque réclamant une attention particulière.

[bookmark: _Toc42955357]QU’EST-CE QU’UNE FORME (SHAPE) POUR EXCEL

Pour EXCEL, la classe « Shape » (traduite ici par « Forme ») comprend une multitude d’objets graphiques.
Dans la figure ci-dessous, quelques objets graphiques membres de la classe « Forme » :
[image: ]
Une forme (shape) peut être un objet « simple » comme une forme « automatique », une icône, une image ou bien un objet « complexe » comme un graphique, un objet SmartArt, un modèle 3D…
Une partie des méthodes et propriétés de la classe « Shape » décrites dans ce document sont utilisables pour toutes les formes.
(Cependant, ayons en tête que certaines propriétés et méthodes de la classe « Shape » ne concernent que certaines d’entre elles, par exemple, voir plus loin la méthode ScaleWidth).


[bookmark: _Toc42955358]UTILISONS L’ENREGISTREUR DE MACROS

 [image: ]Du fait de la richesse et de la diversité des méthodes et propriétés exposées par les formes (shapes), nous gagnons beaucoup de temps en utilisant, dans un premier temps, l’enregistreur de macros. 
Dans un second temps, nous corrigerons le code généré pour l’adapter et/ou le simplifier. 
Pour une bonne utilisation de l’enregistreur de macros, reportons-nous au document : https://www.excel-pratique.com/fr/telechargements/doc-vba/enregistreur-macro-excel-no467


[bookmark: _Toc42955359]LISTONS TOUTES LES FORMES CONTENUES DANS UNE FEUILLE
L’objet ‘Shapes’ d’une feuille de calcul regroupe la collection de toutes les formes (shapes) présentes dans une feuille.
Le fragment de code ci-dessous énumère dans la fenêtre d’exécution les noms de toutes les formes présentes dans la feuille « Feuil1 » de l’exemple :
[image: ]

[image: ][image: ] Pour afficher la fenêtre d’exécution : Ctrl+G. Reportons-nous au document : 
https://www.excel-pratique.com/fr/telechargements/doc-vba/debogage-no469


[bookmark: _Toc42955360]REFERENCONS LES FORMES

[bookmark: _Toc42955361]Référençons par index
Nous pouvons référerencer une forme par son index dans la collection des « Shapes ». Par exemple, pour nous référer à la forme d’index 3 :Set oShape = oSheet.Shapes(3)

Cette méthode de référencement n’est ni très pratique ni très explicite, aussi nous privilégierons le référencement pas nom.

[bookmark: _Toc42955362]Référençons par nom
Lorsque nous créons une forme dans une feuille (manuellement ou par code), EXCEL lui affecte un nom par défaut. Ce nom reprend le type de forme et y ajoute une numérotation chronologique. Par exemple :« Image 7 »

Nous référençons cette forme par nom de la façon suivante :     Set oShape = oSheet.Shapes("Image 7")

Aux noms affectés par défaut par EXCEL, nous pouvons préférer que nos formes portent des noms explicites.

[bookmark: _Toc42955363]Affectons un nom explicite
Nous affectons préférentiellement un nom explicite à nos formes en modifiant la propriété « Name » de la forme.
Par exemple :    Set oShape = oSheet.Shapes("Image 7")
    oShape.Name = "ExcelPratique_Logo"



[bookmark: _Toc42955364]POSITIONNONS LES FORMES

[bookmark: _Toc42955365]Positionnons la forme lors de sa création
Lorsque nous créons une forme dans une feuille par code en utilisant l’une des méthodes disponibles (Addxxx, voir plus loin), nous devons indiquer sa position exprimée en points pour son coin supérieur gauche :
Addxxx (…, gauche, haut, … )
Ainsi, pour ajouter un rectangle biseauté positionné à 5 points à gauche et en haut de la feuille    Set oShape = oSheet.AddShape(msoShapeBevel, 5, 5, …)

[image: ]  Pour convertir les centimètres en points, nous disposons de la fonction ‘Application.CentimetersToPoints’ . Par exemple, pour positionner la forme à 1 centimètre à gauche et 1 centimètre en haut de la feuille nous écrirons : 
Set oShape = oSheet.AddShape(msoShapeBevel, Application.CentimetersToPoints(1), Application.CentimetersToPoints(1), …)

[bookmark: _Toc42955366]Repositionnons une forme
Pour repositionner une forme, nous modifions ses propriétés « Left » et « Top » en indiquant leurs valeurs en points (ou en centimètres en utilisant la fonction ‘Application.CentimetersToPoints’). Par exemple, pour repositionner la forme à 50 points à gauche et en haut de la feuille :oShape.Left = 50
oShape.Top = 50

[bookmark: _Toc42955367]Repositionnons une forme par rapport à une cellule
Nous pouvons positionner une forme par rapport à une cellule en se référençant aux propriétés « Left » et « Top » de la cellule. Par exemple, pour repositionner la forme sur le coin haut-Gauche de la cellule « D4 » :    Set oSheet = ThisWorkbook.Worksheets(1)
    Set oShape = oSheet.Shapes("ExcelPratique_Logo")
    Set oRange = oSheet.Range("D4")

    oShape.Left = oRange.Left
    oShape.Top = oRange.Top

[bookmark: _Toc42955368]Gérons les propriétés « Placement »  
La propriété « Placement » de l’objet « Shape » nous permet de gérer le positionnement des formes.
Elle peut prendre les valeurs :
· xlFreeFloating = la forme conserve son positionnement lorsque les cellules sous-jacentes sont redimensionnées.
· xlMove = la forme se déplace avec les cellules sous-jacentes (sans être redimensionnée).
· xlMoveAndSize = la forme se déplace et se redimensionne avec les cellules sous-jacentes.


[bookmark: _Toc42955369]DIMENSIONNONS LES FORMES

[bookmark: _Toc42955370]Dimensionnons la forme lors de sa création
Lorsque nous créons une forme dans une feuille par code en utilisant l’une des méthodes disponibles (Addxxx, voir plus loin), nous devons indiquer ses dimensions exprimées en points :
Addxxx (…, Largeur, Hauteur, … )
Ainsi, pour ajouter un rectangle biseauté de 100 points sur 100 :    Set oShape = oSheet.AddShape(msoShapeBevel, …, …, 100,100)

[bookmark: _Toc42955371]Redimensionnons une forme 
Pour redimensionner une forme, nous modifions ses propriétés « Width » et « Height » en indiquant leurs valeurs en points (ou en centimètres en utilisation la fonction ‘Application.CentimetersToPoints’).
Par exemple, pour redimensionner la forme à une largeur et une hauteur de 100 points :oShape.Width = 100
oShape.Height = 100

[bookmark: _Toc42955372]Redimensionnons une forme par rapport à la taille d’une cellule
Nous pouvons redimensionner une forme par rapport à la taille d’une cellule (ou d’une plage de cellules) en se référençant aux propriétés « Width » et « Height » de la cellule (ou de la plage).
 Par exemple, pour redimensionner la forme à la taille de la plage « D4:F6 » :    Set oSheet = ThisWorkbook.Worksheets(1)
    Set oShape = oSheet.Shapes("ExcelPratique_Logo")
    Set oRange = oSheet.Range("D4:F6")

    oShape.Width = oRange.Width
    oShape.Height = oRange.Height

Gérons les propriétés « LockAspectRatio »  
La propriété « LockAspectRatio » de l’objet « Shape » nous permet d’indiquer comment la forme doit se comporter lors d’un redimensionnement.
Elle peut prendre les valeurs suivantes :
· msoTrue = la forme se redimensionne en conservant ses proportions. Si, par exemple, nous redimensionnons la largeur de la forme, sa hauteur se redimensionne proportionnellement.
· msoFalse = la forme ne conserve pas ses proportions lorsque nous la redimensionnons. Si, par exemple, nous redimensionnons la largeur de la forme, sa hauteur reste inchangée.

[bookmark: _Toc42955373]Redimensionnons un ensemble de formes avec les méthodes ‘ScaleWidth-ScaleHeight’ 

Avec les méthodes « ScaleWidth » et « ScaleHeight » nous diminuons ou augmentons proportionnellement la taille des formes en largeur (ScaleWidth) ou en hauteur (ScaleHeight).Sub ShrinkAllShapes()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    
    Set oSheet = ThisWorkbook.Worksheets(1)
    'On parcourt toutes les formes présentes dans la feuille
    For Each oShape In oSheet.Shapes
        'On conserve les proportions de la forme
        oShape.LockAspectRatio = msoTrue
        'Dans le cas où la forme ne supporte pas la méthode ScaleWidth, on ignore l'erreur
        On Error Resume Next
        'On diminue la taille de la forme de moitié = 50%
        oShape.ScaleWidth 0.5, msoTrue
        'On rétablit la gestion des erreurs
        On Error GoTo 0
    Next
End Sub

Ce fragment de code, diminue la taille des formes d’une même feuille de la moitié de leurs tailles actuelles :


[image: ]  Certaines formes ne supportent pas la méthode ‘ScaleWidth’ utilisée, c’est la raison pour laquelle l’instruction ‘On Error Resume Next’ est utilisée dans ce fragment de code.
Pour la gestion des erreurs d’exécution, reportons-nous au document : https://www.excel-pratique.com/fr/telechargements/doc-vba/gestion-erreurs-vba-no473)





[bookmark: _Toc42955374]INSERONS UNE NOUVELLE FORME (SHAPE)

L’objet ‘Shapes’ expose un ensemble de méthodes permettant d’insérer par code une nouvelle forme dans une feuille.
Chacune de ces méthodes insère un type de forme (shape) particulier.

AddChart2 (AddChart) – pour ajouter un graphe
(Raccourci Alt+S+….)
[image: ]
La méthode AddChart2 a été introduite avec la version 2013 d’EXCEL. La méthode AddChart a été, quant à elle, introduite dans la version 2007 d’EXCEL.
La différence entre ces deux méthodes tient au fait qu’Addchart2 dispose de 2 arguments supplémentaires :
AddChart (XlChartType, gauche, haut, largeur, hauteur)
AddChart2 (Style, XlChartType, gauche, haut, largeur, hauteur, NewLayout)
Pour créer complètement un graphique par code VBA, nous devrons valoriser des propriétés n’appartenant pas à l’objet ‘Shape’ mais à l’objet ‘Chart’ (par exemple, les données devant composer le graphe). 
Ce fragment de code illustre l’articulation entre l’objet ‘Shape’ et l’objet ‘Chart’ :Sub AddGraph()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    Dim oChart As Chart    
    Set oSheet = ThisWorkbook.Worksheets(2)    
    'On insère le graphe dans la feuille avec la méthode AddChart2
    Set oShape = oSheet.Shapes.AddChart2(333, xl3DPie, , , , True)
    With oShape
        'On référence l'objet Chart
        Set oChart = oShape.Chart
        With oChart
            'On affecte la source de données du graphe
            .SetSourceData oSheet.Range("P3:P12")
            'On donne un titre au graphe
            .ChartTitle.Text = "MON GRAPHE"
        End With
    End With
End Sub

(Ce document est consacré à l’objet formes ‘Shapes’, il ne traite pas en détail de l’objet ‘Chart’)

[bookmark: _Toc42955376]AddCallout – pour ajouter une légende 
 (Raccourci Alt+S+ZI+ER…) :
[image: ]


Ce fragment de code insère une légende avec lien à droite :Sub AddLegend()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    
    Set oSheet = ThisWorkbook.Worksheets(1)
    Set oShape = oSheet.Shapes.AddCallout(msoCalloutOne, 50, 50, 80, 100)    
End Sub

Résultat obtenu :

[image: ]



[image: ]
Dans l’énumération des types exposés par la méthode ‘AddCallout’ la valeur ‘msoCallMixed’ n’est pas fonctionnelle. 



AddConnector – pour ajouter un trait ou un connecteur
(Raccourci Alt+S+ZI+ER….) 

[image: ]

Ce fragment de code insère un trait :Sub AddTrait()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    
    Set oSheet = ThisWorkbook.Worksheets(1)
    Set oShape = oSheet.Shapes.AddConnector(msoConnectorStraight, 50, 50, 100, 100)
End Sub

Résultat obtenu :

[image: ]


[image: ]
La méthode ‘AddConnector’ permet également de relier deux formes entre-elles (voir plus loin dans ce document). 

AddCurve – pour ajouter une courbe de Bézier
(Raccourci Alt+S+ZI+ER….) 
[image: ]
Les points de la courbe doivent être fournis à la méthode ‘AddCurve’ dans un tableau de valeurs. 

Par exemple, ce fragment de code insère une courbe en S avec des coordonnées de début (210,130) et de fin (315,225) :Sub AddCourbeBezier()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    Dim aPoints(1 To 7, 1 To 2) As Single
    aPoints(1, 1) = 210
    aPoints(1, 2) = 130
    aPoints(2, 1) = 270
    aPoints(2, 2) = 150
    aPoints(3, 1) = 280
    aPoints(3, 2) = 160
    aPoints(4, 1) = 280
    aPoints(4, 2) = 175
    aPoints(5, 1) = 200
    aPoints(5, 2) = 190
    aPoints(6, 1) = 210
    aPoints(6, 2) = 205
    aPoints(7, 1) = 315
    aPoints(7, 2) = 225
    
    Set oSheet = ThisWorkbook.Worksheets(1)
    Set oShape = oSheet.Shapes.AddCurve(aPoints())
End Sub

Résultat obtenu :
[image: ]



AddFormControl – pour ajouter un contrôle de formulaire
(Raccourci Alt+V+B….)

[image: ]


Avec la méthode ‘AddFormControl’, nous indiquons le type de contrôle à insérer par la précisant par l’une des valeurs exposées par l’énumération XlFormControl. 

Ce fragment de code insère un contrôle de liste déroulante :Sub AddCombobox()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    
    Set oSheet = ThisWorkbook.Worksheets(1)
    Set oShape = oSheet.Shapes.AddFormControl(xlDropDown, 5, 5, 55, 15)
End Sub

Résultat obtenu : 

[image: ]



AddLabel/AddTextbox – pour ajouter une zone de texte
(Raccourci Alt+S+8….)

[image: ]

Les méthodes ‘AddTextBox’ et ‘AddLabel’, permettent d’insérer une zone de texte ou de libellé.

Ce fragment de code insère une zone de texte et une zone de libellé :Sub AddTextBox_Label()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    
    Set oSheet = ThisWorkbook.Worksheets(1)
    Set oShape = oSheet.Shapes.AddTextBox(msoTextOrientationHorizontal, 5, 5, 200, 15)
    oShape.TextEffect.Text = "Forme ajoutée par AddTextbox"
    Set oShape = oSheet.Shapes.AddLabel(msoTextOrientationHorizontal, 5, 30, 200, 15)
    oShape.TextEffect.Text = "Forme ajoutée par AddLabel"
End Sub

Résultat obtenu : 

[image: ]





AddOLEObject – pour ajouter un lien vers un document externe

· Ajoute un lien à un document externe (Word, PDF, Excel, Texte…) 
Raccourci Alt+S+ZT+W) :

[image: ]


Ce fragment de code insère un lien sous forme d’une icône vers un document PDF intégré au classeur :Sub AddPDF()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    Dim oRange As Range
    
    Set oSheet = ThisWorkbook.Worksheets(1)
    Set oShape = oSheet.Shapes.AddOLEObject(, ThisWorkbook.Path & "\MonPDF.pdf", , , , , , 5, 5, 200, 200)

End Sub

Résultat obtenu : 



[image: ]


AddPicture2/AddPicture – pour ajouter une image
(Raccourci Alt+S+ZI+I1…) 
[image: ]

La seule différence entre les 2 méthodes : la méthode AddPicture2 expose le paramètre supplémentaire « compress » (indiquant si l’image doit être compressée ou non).
[image: ]

Pour convertir les centimètres en points, nous disposons de la fonction ‘Application.CentimetersToPoints’ . 

Ce fragment de code insère le logo « EXCEL Pratique » compressé positionné à 5 points du haut et de la gauche de la feuille, sur une longueur de 10 cms et une largeur de 2 cms :Sub AddImage()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    Dim oRange As Range
    
    Set oSheet = ThisWorkbook.Worksheets(1)
    Set oShape = oSheet.Shapes.AddPicture2(ThisWorkbook.Path & "\Logo_EXCEL_PRATIQUE.jpg", msoFalse, msoTrue, 5, 5 _
                        , Application.CentimetersToPoints(10), Application.CentimetersToPoints(2), True)
End Sub

Résultat obtenu : 
[image: ]


AddPolyline – pour ajouter un polygone ou une ligne brisée
(Raccourci Alt+S+ZI+ER…) 
[image: ]

Les points de la figure voulu sont à fournie à la méthode ‘AddPolyLine’ dans un tableau de valeurs. 

Pour constituer une figure fermée, le dernier point fourni doit posséder les mêmes coordonnées que le premier point fourni.





Ce fragment de code trace un pentagone (figure de 5 côtés).

Sub TracePentagone()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    Dim oRange As Range
    Dim aPoints(1 To 6, 1 To 2) As Single
    
    'On affecte les coordonnées des sommets du pentagone
    aPoints(1, 1) = 520
    aPoints(1, 2) = 93
    aPoints(2, 1) = 433
    aPoints(2, 2) = 211
    aPoints(3, 1) = 180
    aPoints(3, 2) = 194
    aPoints(4, 1) = 70
    aPoints(4, 2) = 76
    aPoints(5, 1) = 232
    aPoints(5, 2) = 51
    aPoints(6, 1) = 520 'le dernier sommet possède des coordonnées identiques au premier sommet
    aPoints(6, 2) = 93  '-> pour fermer le polygone
    
    Set oSheet = ThisWorkbook.Worksheets(2)
    'On insère le pentagone dans la feuille
    Set oShape = oSheet.Shapes.AddPolyline(aPoints)
End Sub

Résultat obtenu :


[image: ]



AddShape – pour ajouter une forme automatique
(Raccourci Alt+S+ZI+ER…) 
[image: ]
Le type de la forme désirée est indiqué par l’argument « Type » de la méthode AddShape (voir l’énumération MsoAutoShapeType  comprenant plus de 180 formes différentes ).

Ce fragment de code insère une forme de type « Bouton en relief » :Sub AddFormeBasique()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    Dim oRange As Range
    
    Set oSheet = ThisWorkbook.Worksheets(3)
    Set oShape = oSheet.Shapes.AddShape(msoShapeBevel, 5, 5, 50, 50)
End Sub


Résultat obtenu :
[image: ]


AddSmartArt – pour ajouter une forme SmartArt
(Raccourci Alt+S+ZI+RM…) 
[image: ]



Pour obtenir la forme voulue, nous spécifions l’objet « SmartArtLayout » voulu comme argument de la méthode « AddSmartArt ».


Le fragment de code suivant, insère la forme SmartArt « Entonnoir » :Sub AddCone()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    Dim oSaLayout As SmartArtLayout
    
    Set oSheet = ThisWorkbook.Worksheets(1)
    'On pointe sur l'illustration 'Entonnoir'
    Set oSaLayout = Application.SmartArtLayouts(68)
    'On créé une nouvelle forme contenant l'illustration voulue
    Set oShape = oSheet.Shapes.AddSmartArt(oSaLayout, 5, 5)
    'On affecte une couleur à la forme
    oShape.SmartArt.Color = Excel.Application.SmartArtColors(4)
End Sub



Résultat obtenu :

[image: ]





[image: ]Ce document traite uniquement de l’objet « Shape ». 
  Pour une référence complète aux formes SmartArt, reportons-nous au document : https://www.excel-pratique.com/fr/telechargements/doc-excel/smartart-reference-no461.
Pour un exemple de création d’organigrammes, reportons-nous au document : https://www.excel-pratique.com/fr/telechargements/gestion-du-personnel/smartart-organigram-no462




AddTextEffect – pour ajouter une forme WordArt
(Raccourci Alt+S+ZT+1) 

[image: ]
Le fragment de code suivant, insère une forme WordArt « EXCEL Pratique » avec effet 7 :Sub AddWordArt()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    
    Set oSheet = ThisWorkbook.Worksheets(1)
    Set oShape = oSheet.Shapes.AddTextEffect(msoTextEffect7, "EXCEL Pratique", "Arial Black", 36 _
                        , msoFalse, msoFalse, 5, 5)
End Sub


Résultat obtenu :

[image: ]







Add3DModel – pour ajouter un modèle 3D 
(Raccourci Alt+S+ZI+S3) 

[image: ]
Le fragment de code suivant, insère une forme 3D animée « Volcan » :Sub Add3D_Volcan()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    
    Set oSheet = ThisWorkbook.Worksheets(1)
    'On insère la forme 3D
    Set oShape = oSheet.Shapes.Add3DModel( _
        "https://cdn.hubblecontent.osi.office.net/models/publish/524bad6218044d5c8e13f4fa93200128/glb/divergent_boundary.glb" _
        , msoFalse, msoTrue, 5, 5, 350, 125)
End Sub


Résultat obtenu :
[image: ]




[bookmark: _Toc42955388]AGISSONS SUR LES COULEURS ET LE REMPLISSAGE 

Les formes (shapes) exposent un ensemble de propriétés nous permettant d’agir sur leurs couleurs et remplissage.
Reprenons le pentagone créé précédemment. 
Le fragment de code suivant, colorise les bordures de ce pentagone par des tirets rouges :Sub ColoreBorduresPentagone()
    Dim oSheet As Worksheet
    Dim oShape As Shape

    Set oSheet = ThisWorkbook.Worksheets(1)
    'On insère la forme 3D
    Set oShape = oSheet.Shapes("Pentagone")
    
    With oShape
        'On colorise les bordures du pentagone en tirets rouges
        With .Line
            .DashStyle = msoLineDashDotDot
            .ForeColor.RGB = RGB(255, 0, 0)
        End With
    End With
End Sub

Le fragment de code suivant, remplit le pentagone de briques vertes :Sub RemplissagePentagone()
    Dim oSheet As Worksheet
    Dim oShape As Shape

    Set oSheet = ThisWorkbook.Worksheets(1)
    'On insère la forme 3D
    Set oShape = oSheet.Shapes("Pentagone")
    
    With oShape
        'On remplit le pentagone de briques vertes
        With .Fill
            .Patterned msoPatternHorizontalBrick
            .BackColor.RGB = RGB(0, 255, 0)
        End With
    End With
End Sub


Résultat obtenu :

[image: ]





[bookmark: _Toc42955389]AJOUTONS ET FORMATONS DU TEXTE 

Toutes les formes (shapes) exposent une propriété ‘TextFrame’ nous permettant d’ajouter du texte et de le mettre en forme.

Reprenons la forme « Légende » créée précédemment et ajoutons du texte avec le fragment de code :Sub AddTextToLegend()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    
    Set oSheet = ThisWorkbook.Worksheets(1)
    Set oShape = oSheet.Shapes("Legende")
    oShape.TextFrame.Characters.Text = "EXCEL-PRATIQUE : c'est la top ressource !"
End Sub

Résultat obtenu :

[image: ]



Formatons le texte inséré précédemment avec le fragment de code :Sub FormatTextToLegend()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    
    Set oSheet = ThisWorkbook.Worksheets(1)
    Set oShape = oSheet.Shapes("Legende")
    
    'Formatons les 16 premiers caractères du texte : corps 12, en gras et en vert
    With oShape.TextFrame.Characters(1, 16).Font
        .Color = RGB(0, 255, 0)
        .Size = 12
        .Bold = True
    End With
    'Formatons les 3 caractères à la partir de la position 27 du texte: corps 16, en gras, en rouge et soulignés
    With oShape.TextFrame.Characters(27, 3).Font
        .Color = RGB(255, 0, 0)
        .Size = 16
        .Bold = True
        .Underline = xlUnderlineStyleSingle
    End With
    'Centrons le texte horizontalement et verticalement
    oShape.TextFrame.HorizontalAlignment = xlHAlignCenter
    oShape.TextFrame.VerticalAlignment = xlVAlignCenter
End Sub

Résultat obtenu :

[image: ]


[bookmark: _Toc42955390]ASSIGNONS UNE MACRO

Notre possibilité d’assigner une macro à une forme (Shape) constitue une caractéristique particulièrement intéressante.
Soit la macro (simplissime) suivante :Sub displayInfo()
    MsgBox "Lien vers EXCEL-Pratique: " & vbCrLf _
         & "https://www.excel-pratique.com/" _
         , vbExclamation, "MESSAGE D'INFORMATION"
End Sub

Affectons cette macro à la forme « Légende » créée précédemment avec le fragment de code :Sub displayInfo()
    MsgBox "Lien vers EXCEL-Pratique: " & vbCrLf _
         & "https://www.excel-pratique.com/" _
         , vbExclamation, "MESSAGE D'INFORMATION"
End Sub


Résultat obtenu lors d’un clic sur « Légende » :
[image: ]



RELIONS DES FORMES AVEC UN CONNECTEUR

Voyons comment nous pouvons relier deux formes (shapes) entre-elles au moyen d’une forme « Connecteur ».
Commençons par tracer un ordinogramme comportant 4 formes au moyen du code suivant :



Résultat obtenu :
[image: ]


Pour relier 2 formes entres elles, créons une forme ‘Connecteur’ au moyen de la méthode « AddConnector » (voir plus haut) puis utilisons les méthodes de l’objet ‘ConnectorFormat’ : 
       BeginConnect et EndConnect

Par exemple, au moyen du code suivant, nous relions les formes de notre ordinogramme deux à deux :


Résultat obtenu :
[image: ]



[image: ]
Les méthodes BeginConnect et EndConnect réclament le paramètre obligatoire ‘SiteConnection’ indiquant le point d’ancrage du connecteur sur la forme à relier.
Ce point d’ancrage est à choisir parmi les points d’ancrage existants pour la forme à relier. Chaque type de forme possède sa propre collection de points d’ancrage.
Du fait qu’EXCEL n’expose pas la collection des points d’ancrage d’une forme donnée, il n’est pas aisé de déterminer le point nous convenant pour le tracé désiré.
Si, comme dans notre exemple, nous désirons positionner les connecteurs à des points bien spécifiques des formes, le plus simple consiste à utiliser l’enregistreur de macro afin de déterminer les numéros de ces points.
Une autre option consiste à indiquer le point d’ancrage ‘1’ et d’utiliser la méthode ‘RerouteConnections’ de l’objet ‘Shape’ qui a pour effet de prendre le chemin le plus court entre les deux formes reliées. Par exemple :
    Set oConnector = zSheet.Shapes.AddConnector(msoConnectorElbow, 120, 50, 200, 100)       
    With oConnector
        'Relions les 2 formes avec le connecteur
        .ConnectorFormat.BeginConnect zShape1, 1
        .ConnectorFormat.EndConnect zShape2, 1
        .RerouteConnections
    End With


BONNES PRATIQUES VBA : Du bon usage de l’enregistreur de macro		1

Option Explicit

Sub InventaireFormes ()
Din oSheet As Worksheet
Dim oShape As Shape

Set oSheet = ThisWorkbook.Worksheets (1)

For Each oShape In oSheet.Shapes
Debug.Print oShape.Name
Next
End Sub

Add4ShapesCode.txt


Add4ShapesCode.txt
Sub Add4Shapes()

    Dim oSheet As Worksheet

    Dim oShape As Shape

    Dim oRange As Range

    

    Set oSheet = ThisWorkbook.Worksheets(1)

    

    'Insérons la forme 'Database'

    InsertNewShape oSheet, msoShapeFlowchartMagneticDisk, "BaseSQL", "B2:B6", msoThemeColorAccent4, "Base SQL"

    

    'Insérons la forme 'Traitement'

    InsertNewShape oSheet, msoShapeFlowchartProcess, "Traitement", "C9:D12", msoThemeColorAccent6, "Traitement"

    

    'Insérons la forme 'Etats'

    InsertNewShape oSheet, msoShapeFlowchartMultidocument, "Etats", "B16:B19", msoThemeColorAccent5, "Etats 1,2,3..."

    

    'Insérons la forme 'Ecran'

    InsertNewShape oSheet, msoShapeFlowchartDisplay, "Display", "E16:E19", msoThemeColorAccent5, "Ecran"

End Sub

Sub InsertNewShape(zSheet As Worksheet, zType As Variant, zName As String, zRange As String, zColor As Variant, zText As String)

    Dim oShape As Shape

    Dim oRange As Range

    

    'Insérons la forme

    Set oShape = zSheet.Shapes.AddShape(zType, 5, 5, 40, 70)

    With oShape

        'Donnons un nom à la forme

        .Name = zName

        'Positionnons et dimensionnons la forme

        Set oRange = zSheet.Range(zRange)

        .Top = oRange.Top

        .Left = oRange.Left

        .Height = oRange.Height

        .Width = oRange.Width

        'Colorisons la forme

        .Fill.ForeColor.ObjectThemeColor = zColor

        .Fill.ForeColor.Brightness = 0.6

        'Ajoutons du texte dans la forme

        .TextFrame2.TextRange.Characters.Text = zText

        'Colorisons et formatons le texte

        With .TextFrame2.TextRange.Characters.Font

            .Bold = True

            .Fill.ForeColor.RGB = RGB(0, 0, 0)

        End With

        'Centrons le texte dans la forme

        With .TextFrame2

            .TextRange.Characters.ParagraphFormat.Alignment = msoAlignCenter

            .VerticalAnchor = msoAnchorMiddle

        End With       

    End With

End Sub

AddConnectorsCode.txt
Sub AddConnectors()

    Dim oSheet As Worksheet

    Dim oShape1 As Shape, oShape2 As Shape

    Dim oRange As Range

    

    Set oSheet = ThisWorkbook.Worksheets(1)

    

    'Connectons la forme 'Database' à la forme "Traitement"

    Set oShape1 = oSheet.Shapes("BaseSQL")

    Set oShape2 = oSheet.Shapes("Traitement")

    InsertConnector oSheet, oShape1, oShape2, 5, 1



    'Connectons la forme 'Traitement' à la forme "Etats"

    Set oShape1 = oSheet.Shapes("Traitement")

    Set oShape2 = oSheet.Shapes("Etats")

    InsertConnector oSheet, oShape1, oShape2, 3, 1



    'Connectons la forme 'Traitement' à la forme "Display"

    Set oShape1 = oSheet.Shapes("Traitement")

    Set oShape2 = oSheet.Shapes("Display")

    InsertConnector oSheet, oShape1, oShape2, 3, 1



End Sub

Sub InsertConnector(zSheet As Worksheet, zShape1 As Shape, zShape2 As Shape, zPoint1 As Long, zPoint2 As Long)

    Dim oConnector As Shape

    Dim oRange As Range

    

    'Insérons le connecteur

    Set oConnector = zSheet.Shapes.AddConnector(msoConnectorElbow, 120, 50, 200, 100)

    

    

    With oConnector

        'Formatons le connecteur

        .Line.Weight = 1.75

        .Line.ForeColor.ObjectThemeColor = msoThemeColorAccent1

        .Line.EndArrowheadStyle = msoArrowheadTriangle

        'Relions les 2 formes avec le connecteur

        .ConnectorFormat.BeginConnect zShape1, zPoint1

        .ConnectorFormat.EndConnect zShape2, zPoint2

        '.RerouteConnections

    End With

End Sub

